IREM

INSTITUT DE RECHERCHE SUR L'ENSEIGNEMENT DES MATHÉMATIQUES
DES PAYS DE LA LOIRE



Logo et géométrie



En 4^e – 3^e

2007

Alain BOIS – Jacques DELGOULET

Logo
et
géométrie

Alain BOIS
Jacques DELGOULET

Logo
et
géométrie

En 4^{e} – 3^{e}

IREM

Présentation générale .... page 1

Descriptions des macro-procédures .... page 2

Compte rendu d'expérimentations en classe de Quatrième et Troisième (J. Delgoulet Collège A.Fournier) . page 12

Utilisation de macro-procédures en Géométrie ..... page 33

Tracés de coniques .... page 36

Recherches de barycentres .... page 40

Ecrire une brochure IREM n'est jamais facile mais cette fois-ci ,j'ai l'impression que cela sent le "réchauffé" (nous en avons écrit la majeure partie en 1986) . Mais tant pis ,j'ai la vanité de croire qu'elle peut rendre service à des collègues qui n'ont jamais osé utiliser LOGO dans leurs classes, c'est pourquoi j'ai ajouté quelques pages d'exemples d'utilisation.

Cette brochure peut être divisée en trois parties :

- une étude théorique qui est le résumé des stages PAF-IREM intitulés «LOGO et macro-procédures en Géomètrie»

- un résumé des actions pédagogiques réalisées par J.Delgoulet au collège A.Fournier du Mans .

- des idées d'utilisation du "logiciel" :
  - tracés des bissectrices d'un triangle (mode direct et mode procédural)
  - tracés de coniques
  - recherche de barycentres

Remarques concernant la première partie :

En 1985,les premiers TO7 apparaissaient dans la Sarthe, nous avions déjà des idées de "logiciels" sur APPLE ,nous désirions transférer ces derniers sur le matériel qui nous était fourni mais nous voulions aussi profiter (!!!) des nouvelles possibilités offertes par le stylo optique (nous avons déchanté depuis) .

Je vais donc essayer de relater le plus fidèlement possible les problèmes que nous nous sommes posés et la méthode utilisée pour les résoudre . Les solutions proposées ne sont pas toujours "académiques" tant sur le plan mathématique qu'informatique . J'ai eu le désir de former mes collègues à la programmation LOGO dans l'esprit LOGO , c'est à dire en partant des acquis sensoriels de géomètrie : pour certaines solutions je leur proposais de se mettre à la place de la tortue . C'est pourquoi les solutions purement analytiques n'ont été écrites qu'à titre d'information et n'ont jamais été privilégiées dans ce stage .

Voici le plan de la première partie :

1. Le tracé du carré au stylo optique
2. Le tracé du rectangle au stylo optique
3. Le tracé de lignes caractéristiques du triangle
4. L'intersection de 2 droites
5. L'intersection d'un cercle et d'une droite
6. L'intersection de 2 cercles

Une disquette contenant ces macro-procédures est en vente au CARI de Nantes .

Présentation générale .... page 1

Descriptions des macro-procédures .... page 2

Compte rendu d'expérimentations en classe de Quatrième et Troisième (J. Delgoulet Collège A.Fournier) . page 12

Utilisation de macro-procédures en Géométrie ..... page 33

Tracés de coniques .... page 36

Recherches de barycentres .... page 40

Ecrire une brochure IREM n'est jamais facile mais cette fois-ci ,j'ai l'impression que cela sent le "réchauffé" (nous en avons écrit la majeure partie en 1986) . Mais tant pis ,j'ai la vanité de croire qu'elle peut rendre service à des collègues qui n'ont jamais osé utiliser LOGO dans leurs classes, c'est pourquoi j'ai ajouté quelques pages d'exemples d'utilisation.

Cette brochure peut être divisée en trois parties :

- une étude théorique qui est le résumé des stages PAF-IREM intitulés «LOGO et macro-procédures en Géomètrie»

- un résumé des actions pédagogiques réalisées par J.Delgoulet au collège A.Fournier du Mans .

- des idées d'utilisation du "logiciel" :
  - tracés des bissectrices d'un triangle (mode direct et mode procédural)
  - tracés de coniques
  - recherche de barycentres

Remarques concernant la première partie :

En 1985,les premiers TO7 apparaissaient dans la Sarthe, nous avions déjà des idées de "logiciels" sur APPLE ,nous désirions transférer ces derniers sur le matériel qui nous était fourni mais nous voulions aussi profiter (!!!) des nouvelles possibilités offertes par le stylo optique (nous avons déchanté depuis) .

Je vais donc essayer de relater le plus fidèlement possible les problèmes que nous nous sommes posés et la méthode utilisée pour les résoudre . Les solutions proposées ne sont pas toujours "académiques" tant sur le plan mathématique qu'informatique . J'ai eu le désir de former mes collègues à la programmation LOGO dans l'esprit LOGO , c'est à dire en partant des acquis sensoriels de géomètrie : pour certaines solutions je leur proposais de se mettre à la place de la tortue . C'est pourquoi les solutions purement analytiques n'ont été écrites qu'à titre d'information et n'ont jamais été privilégiées dans ce stage .

Voici le plan de la première partie :

1. Le tracé du carré au stylo optique
2. Le tracé du rectangle au stylo optique
3. Le tracé de lignes caractéristiques du triangle
4. L'intersection de 2 droites
5. L'intersection d'un cercle et d'une droite
6. L'intersection de 2 cercles

'ne disquette contenant ces macro-procédures est en vente au CARI de Nantes .

1. PROBLEME DU TRACE DU CARRE:

Il faut bien avoir à l'idée que nous cherchons une solution non analytique ne faisant pas appel au primitif VERS .

Définition du problème : L'ordinateur trace un carré dont deux sommets opposés ont été pointés au stylo optique par un élève .

Solution possible (voir fig.1)

Questions : - où se trouve B ?
- lorsque B est trouvé, comment trouver D ?
- comment joindre les 4 points ?

Pour trouver le point B, on trace le cercle de centre M et de diamètre AC .

C'est alors qu'intervient un problème typiquement lié au fait que l'écran d'un ordinateur n'est pas un plan continu mais est constitué de points qui occupent une surface non négligeable.

Existe t-il un point du cercle qui se trouve sur la médiatrice de [AC] ? Peu de chance .

Par contre il existe ce que j'appelle un «point de rupture », c'est le moment où un point du cercle passe de la gauche à la droite de la médiatrice .

Comment reconnaître cet état ?

Le point m_1 est tel que $m_1A - m_1B < 0$ ‖
Le point m_2 est tel que $m_2A - m_2B > 0$ ‖ voir fig 2.

Pour trouver le point B, il suffit en première approximation de prendre le milieu de $[M_nM_{n+1}]$



FIG. 1



Fig. 2

Essayons de traduire cette démarche plus physique que mathématique en LOGO.

Quels sont les outils dont nous avons besoin ?

1. Milieu de 2 points
2. Cercle dont on connait le centre et le diamètre .
3. Tracé d'une ligne polygonale

```
POUR MILIEU :A :B
RENDS PH DEMI.SOMME PREM :A PREM :B DEMI.SOMME DER :A  DER :B
FIN

          POUR DEMI.SOMME :A :B
          RENDS DIV SOMME :A :B 2
          FIN

POUR CERCLE
DONNE "Rayon DISTANCE :A MILIEU :A :B
LC AV :R TD 93
CHERCHEB
FIN

          POUR DISTANCE :A :B
          RENDS RC SOMME CARRE DIFF PREM :A PREM :B
                         CARRE DIFF DER :A DER :B
          FIN
                                        POUR CARRE :A
                                        RENDS :A * :A
                                        FIN
POUR CHERCHEB
DONNE "L1 DIFF DISTANCE POS :A POS :B
DONNE "M1 POS
AV 2 * 3.14 * :R / 60 TD 6
DONNE "M2 POS
DONNE "L2 DIFF DISTANCE POS :A POS :B
SI PLP? :L1 * :L2 0 [ DONNE "B MILIEU :M1 :M2 STOP ]
CHERCHEB
FIN
```

Nous avons besoin d´une procédure qui nous fournit les coordonnées du symétrique d´un point par rapport à un autre .

```
POUR SYM :A :B
RENDS PH DIFF 2 * PREM :B PREM :A
         DIFF 2 * DER :B DER :A
FIN
```

Décrivons cinématiquement la recherche du tracé du carré .

```
POUR Carré
Positionne.A.et.C
LC FPOS MILIEU :A :C
CERCLE
LC FPOS :B
BC FPOS PH :C PH SYM :B PH MILIEU :A :C :A :B
FIN

POUR Positionne.A.et.C
EC [ Pointez le stylo sur les extrémités d´une diagoanale ]
DONNE "A STYLO
ATTENTE 100
DONNE "C STYLO
FIN

POUR ATTENTE :A
SI :A = 0 [ STOP ] [ATTENTE :A - 1 ]
FIN

POUR STYLO
SI CONTACT? [ RENDS POSOPT ][ RENDS STYLO ]
FIN
```

La procédure STYLO est élèmentaire et a été améliorée par la suite .

Très rapidement, nous nous sommes aperçus que la primitive VERS qui était présente sur le LOGO APPLE nous serait bien utile et nous allons montrer comment l'écriture précédente lourde pouvait s'alléger (fig.3) .

Démarche géomètrique :

1. la tortue se place en A.
2. elle se tourne vers C
3. elle avance de AC/2
4. elle tourne de 90 degrés
5. elle avance de AC/ 2
6. on récupére la position de B.
7. on recule de AC, on récupère D
8. on trace ABCD .



FIG.3

Ce qui donne en LOGO :

```
POUR Carré
PositionneA.et.C
LC FPOS :A FCAP VERS :C AV DIV DISTANCE :A :C  2
TD 90 AV DIV DISTANCE :A :C 2
DONNE "D POS
RE DISTANCE :A :C
DONNE "B POS
LC FPOS :A
BC FPOS PH :B PH :C PH :D :A
FIN
```

Un peu plus élégant non !!!!

2. PROBLEME DU TRACE DU RECTANGLE :

Définition du problème : L'élève pointe son stylo optique sur 2 sommets opposés du rectangle et sur un point d'un des côtés du rectangle. (Voir fig. 4) .

Solution possible (sans le primitif VERS) :

B se trouve sur le cercle de diamètre AC et sur la droite AM.

Sous-problème: Comment savoir si un point du cercle se trouve sur la droite AM ?

Idée : Cherchons la colinéarité des vecteurs $\overrightarrow{AM}$ et $\overrightarrow{AB}$.
Attention, il y a peu de chance de trouver un point ayant cette propriété (voir remarque sur le carré) . Mais par contre, on peut peut-être trouver encore un « changement d'état ».

Nous allons chercher ce qui change lorsqu'un point du cercle passe de « la gauche à la droite » de B (Voir fig. 5) .



o points placés par le stylo optique

• point mobile

□ point cherché



FIG.4

FIG.5

Le déterminant des vecteurs $\overrightarrow{AM}$ et $\overrightarrow{AB_n}$ change de signe .

Ce qui en LOGO va donner :

```
POUR CHERCHEB
DONNE "M1 POS
DONNE "det1 DET :A :M :A :M1
SI :det1 = 0 [ DONNE "B POS STOP ]
AV 2 * 3.14 * :R / 60 TD 6
DONNE "M2 POS
DONNE "det2 DET :A :M :A :M2
  SI :det2 = 0 [ DONNE "B :M2 STOP ]
  SI PROD :det1 :det2 < 0 [ DONNE "B MILIEU :M1 :M2 STOP ]
CHERCHEB
FIN
```

Fabriquons la procédure DET

```
POUR DET :A :B :C :D
DONNE "xAB DIFF PREM :A PREM :B
DONNE "xCD DIFF PREM :C PREM :D
DONNE "yAB DIFF DER :A DER :B
DONNE "yCD DIFF DER :C DER :D
RENDS DIFF :xAB * :yCD :xCD * :yAB
FIN
```

Comme vous voyez, LOGO permet de travailler dans un esprit proche de ce que l'on « voit » même si c'est au prix de quelques acrobaties qui sont ,je l'avoue, assez loin, d'une approche purement mathématique .

Avec le primitif VERS (fig. 6)

FIG.6



$MB = MC \times \cos\alpha$

D'après la relation de Chasles appliquée aux angles orientés, nous avons :

$$(\widehat{MB, MC}) = (\widehat{MB, \Delta}) + (\widehat{\Delta, MC})$$
$$= -(\widehat{\Delta, AM}) + (\widehat{\Delta, MC})$$

Faisons calculer tout celà à la tortue :

La tortue se place en M, on récupère VERS :C

La tortue se place en A, on récupère VERS :M

EN RESUME

La tortue se place en A et on récupère VERS :M
puis se place en M pour récupèrer VERS :C
ensuite il suffit de fixer le CAP vers M et d'avancer de MC * COS (DIFF VERS :C VERS :M)

EN LOGO ......

```
POUR RECTANGLE
PositionneA.C.et.M
LC FPOS :a DONNE "C1 VERS :M
LC FPOS :M DONNE "C2 VERS :V
FCAP :C1 AV PROD DISTANCE :M :C COS DIFF :C2 :C1
DONNE "B POS .........
```

Ou peut-être dans un esprit plus « tortue » (je n'ai pas dit tortueux !!!)

```
POUR RECTANGLE
Positionne ....
CHERCHEB2
FPOS :C
TD 180 AV DISTANCE :A :B
FPOS :A
FIN

POUR CHERCHEB2
DONNE "C1 VERS :M
BC FPOS :M DONNE "C2 VERS :C
FCAP :C1 AV AV PROD DISTANCE :M :C COS DIFF :C2 :C1
FIN
```

3. TRACES DANS LE TRIANGLE :

31. Tracés des hauteurs :





Bel exemple de récursivité non ?
Ce procédé présente cependant un inconvénient, il « marche » que si les angles à la base sont aigus .
Pour le plaisir (?!) écrivons quand même les procédures .

```
POUR HAUTEUR :A :B :C (hauteur issue de A )
CHERCHEH
LC FPOS :A BC FPOS :H
FIN
```

```
POUR CHERCHEH
DONNE "M MILIEU :B :C
SI PROCHE? ANGLE :A :M :C 90 [ DONNE "H :M STOP ]
TESTE PLP? DISTANCE :A :M DISTANCE :A :C
SIVRAI [ DONNE "C :M ]
SIFAUX [ DONNE "B :M ]
CHERCHEH
FIN
```

Deux procédures sont à construire : PROCHE? et ANGLE ,

```
POUR PROCHE? :A :B
RENDS PLP? ABS DIFF :A :B   1
FIN
```

```
POUR ANGLE :A :B :C
DONNE "POS POS DONNE "CAP CAP
```

(Et oui , c´est bien gentil de promener la tortue pour aller mesurer des angles mais il faudrait la remettre en place à la fin de son travail sinon ne soyez pas surpris de la retrouver à des endroits !!!!)

```
LC FPOS :B DONNE "C1 VERS :A DONNE "C2 VERS :C
FPOS :POS FCAP :CAP
RENDS RESTE ABS DIFF :C1 :C2 180
FIN
```

32. Problème du tracé des bissectrices :

Pour tracer la bissectrice intérieure d´un angle d´un triangle, nous allons nous servir de la propriété suivante (Voir fig. 8)

Le point I est le barycentre des points A et B affectés des coefficients 1 et k avec k = AB / BC

Ce qui donne pour les coordonnées de I :

xI = (xA + k * x C) / (1 + k)
yI = (yA + k * y C) / (1 + k)



Ce qui donne en LOGO :

```
POUR BISSECTRICE :A :B :C
CHERCHEI
LC FPOS :B BC FPOS :I
FIN
```

```
POUR CHERCHEI
DONNE "k DIV DISTANCE :A :B DISTANCE :B :C
DONNE "xI DIV SOMME PREM :A PROD :k PREM :C SOMME 1 :k
DONNE "yI DIV SOMME DER :A PROD :k DER :C SOMME 1 :k
DONNE "I PH :xI :yI
FIN
```

..... et avec VERS ......

```
POUR TRACE.BISSECTRICE :A :B :C
LC FPOS :A FCAP VERS :C
DONNE "k DIV DISTANCE :A :B DISTANCE :B :C
AV PROD DISTANCE :A :C QUOT :k :k + 1
FIN
```

Vous trouverez dans la deuxième partie des procédures relatives aux bissectrices abordables avec des élèves de 4ème .

4. Intersection de 2 droites :

Nous avons d'abord étudié une solution trigonométrique (fig.9).



Fig. 9

Dans le cas de cette figure, nous avons :

$$\overline{AB} = AH / \sin(\overrightarrow{AB}, \overrightarrow{CD})$$

```
POUR INTER :A :B :C :D
( il faudra prévoir le cas où les droites
sont parallèles)
LC FPOS :A
DONNE "d1 DISTANCEDROITE :A :C :D
FCAP VERS :B
AV :D1 / SIN Angle.vecteurs :A :B :C :D
RENDS POS
FIN
```




```
POUR DISTANCEDROITE :A :B :C
RENDS ABS PROD DISTANCE :A :B SIN  Angle.vecteurs :B :A :B :C
FIN
```

```
POUR Angle.vecteurs :A :B :C :D
DONNE "POS POS DONNE "CAP CAP
LC FPOS :C DONNE "C1 VERS :D
FPOS :A DONNE "C2 VERS :B
FPOS :POS FCAP :CAP
RENDS :C2 - :C1
FIN
```



Tout semblait correct sauf que l'on n'avait pas étudié le cas de la figure 10 . Une étude succinte de ce cas nous permit de voir que l'on trouvait dans ce cas le symétrique du point cherché par rapport au point A (il faut plus pour décourager un logoïste) .

Donc, dans INTER nous introduirons le test suivant :

```
TESTE NON PROCHE? DISTANCE.DROITE POS :A :B 0
SIVRAI [ RENDS SYM POS :A ]
SIFAUX [ RENDS POS ]
```

Voyons maintenant d'autres solutions possibles pour trouver l'intersection de 2 droites .

1. solution analytique « pure »

La droite d_1 a pour équation $y = ax + b$

La droite d_1 a pour équation $y = ax + b$

La droite d_2 a pour équation $y = cx + d$

Donc, $x0 = (d - b) / (a - c)$ et $y0 = a . x0 + b$

ATTENTION, il faudra prévoir le cas où les droites sont parallèles et aussi le cas où l'une des droites est parallèle à l'axe des y .

Voyons ce que ça donne comme procédures :

```
POUR INTER :A :B :C :D
   Calcul.de.a
   Calcul.de.b
   Calcul.de.c
   Calcul.de.d
   DONNE "x0 ( :d - :b ) / ( :a - : c)
   RENDS PH :x0 :a * :x0 + :b
FIN

POUR Calcul.de.a
  DONNE "a DIV DIFF DER :B DER :A DIFF PREM :B PREM :A
FIN

POUR Calcul.de.b .....
```

Il serait plus judicieux de fabriquer une procédure COEF.DIR qui rend le coefficient directeur d´une droite qui passe par 2 points

```
POUR COEF.DIR :A :B
  RENDS DIV DIFF DER :B DER :A DIFF PREM :B PREM :A
FIN
```

Dans ce cas, Calcul.de.a devient :

```
POUR Calcul.de.a
   DONNE "a COEF.DIR :A :B
FIN
```

Voyons enfin une solution que je qualifierai d´amusante ou de surprenante pas du tout académique .

Décrivons d´abord l´organigramme :



Les procédures LOGO maintenant :

```
POUR INTER :A :B :C :D
   LC FPOS :A FCAP VERS :B
   RENDS VA.CHERCHER.O
FIN

POUR VA.CHERCHER.O
   SI Tu.es.sur :C :D [RENDS POS ]
DONNE "d DISTANCE.DROITE POS :C :D
   SI NON Tu.es.plus.près.de :C :D [TD 180 AV 2 ]
RENDS VA.CHERCHER.O
FIN

POUR Tu.es.sur :A :B
   RENDS PROCHE? DISTANCE.DROITE POS :C :D 0
FIN

POUR Tu.es.plus.près.de :C :D
   RENDS PLP? DISTANCE.DROITE POS :C :D :d
FIN
```

Les procédures PROCHE? et DISTANCE.DROITE ont déjà été étudiées .

Il nous reste à étudier un problème important qui est celui de l'intersection de 2 cercles et d'un cercle et d'une droite .

5 et 6 . Intersections de cercles :

La démarche suivie n'est pas naturelle mais est efficace , elle met en jeu le centre d'homotéthie de 2 cercles . (voir fig. 10)

Fig. 10



Détaillons les procédures :

1. Il faut trouver le centre d'homotéthie
2. Il faut chercher les points de tangences aux 2 cercles
3. Enfin, il faut écrire une procédure qui donne les points d'intersection d'une droite et d'un cercle .

Allons-y !!!!

```
POUR Centre.d'homotéthie :O :R :O' :R'
  DONNE "POS POS
(Pour remettre la tortue là où on l'a trouvée )
LC FPOS :O FCAP VERS :O'
  AV DIV PROD DISTANCE :O :O' :R DIFF :R :R'
  DONNE "#POS POS FPOS :POS
  RENDS :#POS
FIN          (voir fig. 11)
```

```
POUR TANGENCES :w :O :R
  DONNE "POS POS
  SI PLP? DISTANCE :O :w :R [ EC [Point intérieur] RENDS []]
  AV DIV CARRE :R DISTANCE :O :w
  DONNE "%H POS
  TD 90 AV RC PROD DISTANCE :O :%H DISTANCE :%H :w
  DONNE "#POS POS FPOS :POS
  RENDS LISTE :#POS SYM :#POS :%H
FIN        (Voir fig.12)
```



Fig. 11

Fig 12


Fig. 13


Ecrivons la procédure INTERCERCLES

```
POUR INTERCERCLES :O :R :O´ :R´
   DONNE "%O Centre.d´homotéthie :O :R :O´ :R´
   DONNE "%T PREM TANGENCES :%O :O :R
   DONNE "%T´ PREM TANGENCES :%O :O´:R´
   DONNE "%M MILIEU :%T :%T´
   DONNE "%H .PROJETEORTHO :%M :O :O´
   RENDS INTERDRCE :%M :%H :O :R
FIN
```

Il n´y plus qu´à écrire la procédure INTERDRCE .. (Voir fig.13)

```
POUR INTERDRCE :A :B :O :R
  DONNE "POS POS
  DONNE "%H .PROJETEORTHO :O :A :B
  DONNE "%d RC DIFF CARRE :R CARRE DISTANCE :O :%H
  LC FPOS :%H FCAP VERS :A AV :%d
  DONNE "#POS POS FPOS :POS
  RENDS LISTE :#POS SYM :#POS :%H
FIN
```

Remarques : Vous pouvez être étonné(e) de voir des variables avec des noms #H ,%J .... J´ai été amené à faire celà pour éviter que des variables auxiliaires n´interférent sur des noms de points .

Cette procédure a été améliorée parce qu´elle ne prenait pas en compte le cas où les 2 cercles ont même rayon .

Fin provisoire ..... Juillet 1986

Compte rendu d'expérimentations effectuées

en 4ème au collège Alain Fournier du MANS

Compte rendu d'utilisation en stage IREM

I. PRESENTATION

Le programme GEOMETRIE a été élaboré à partir d'une idée d'Alain Bois qui a construit la majeure pertie des procédures écrites en LOGO .

Le principe de GEOMETRIE consiste à simuler à l'écran la quasi-totalité des figures de géométrie rencontrées en premier cycle (voir même en second cycle) .

Il s'agit d'un logiciel totalement ouvert fonctionnant constamment en mode direct . L'ordinateur n'interroge jamais l'élève pas plus qu'il n'évalue son travail (c'est LOGO qui effectue le contrôle des saisies , la plupart étant des erreurs de syntaxe) . C'est donc à l'élève d'agir sur le système et de résoudre les situations-problèmes proposées à l'aide du supplément de langage fourni .

Le logiciel se présente sous la forme d'une série de procédures permettant de faire des tracés de toutes sortes . Leur nombre est limité et leur syntaxe est la plus simple possible .

Exemples

*** DONNE "I MILIEU :A :B
La variable I contient les coordonnées du milieu du segment [AB] , A et B ayant été préalablement affectées .

*** TRACE [A B] trace le segment [AB]

*** CERCLEC :O DISTANCEDROITE :O :A :B : Trace le cercle de centre O tangent à la droite (AB)

*** DONNE "O INTREDROITES :A :B :C :D La variable O contient les coordonnées du point d'intersection des droites (AB) et (CD) .

GEOMETRIE se présente sur disquettes et cassettes fonctionnant sur MO5, TO7 et Nanoréseau .

Le programme est accompagné de fiches présentant les diverses macro-procédures . Celles qui interviennent le plus fréquemment sont présentées dans la boîte à outils . Les autres présentées dans les fiches G0 , G1 et G2 peuvent être introduites d'un seul coup ou progressivement en fonction des méthodes de chacun .

Il semble malgré tout qu'il est utile de faire « sentir » la nécessité d'un outil avant de le présenter (exemple : après avoir tracé le segment [AB] , le tracé de sa médiatrice peut être l'occasion d'introduire le mot MILIEU) .

Les procédures utilisées sont de natures différentes : certaines effectuent des tracés (TRACE , SEGMENT ...) , d'autres retournent des nombres (DISTANCE , DISTANCEDROITE ..) et d'autres enfin des points (INTERDROITES , PROJETE ...) ou des listes de points (INTERCERCLES) .

La présentation détaillée de ces procédures est faite dans la première partie de cet ouvrage .

II. MISE EN OEUVRE DANS UNE CLASSE

GEOMETRIE a été utilisé dans les trois classes de 4ème du collège A. Fournier , dès le mois de Janvier 86 pour deux d'entre elles, plus tard pour la dernière à raison d'un heure par semaine de façon plus ou moins régulière (environ 4 heures ont été nécessaires pour présenter LOGO d'une part et le logiciel d'autre part) .

Il est important de préciser que le but n'était pas de faire du LOGO ni de l'Informatique mais d'utiliser en laison avec des ac tivites géométriques classiques un nouvel outil .

Le travail s'est fait conjointement à partir des situations rencontrées en classe (voir travaux d'élèves en annexe) et de l'étude des fiches (G0 , G1 et G2) .

Le nombre volontairement limité de procédures et leur syntaxe se rapprochant le plus possible de la réalité ont facilité une appropriation aisée du logiciel par les élèves qui très rapidement ont acquis une dextérité étonnante . La possibilité d'enrichir l'environnement par la création de nouvelles procédures (MEDIATRICE , BISSECTRICE par exemple) a été très motivante pour certains élèves .

Trois directions d'utilisation peuvent être envisagées :

- en classe à partir de thèmes précis (médiatrices , hauteurs , quadrilatère , symétries , projections , Pythagore ....)

- ces activités pour certains , ont été prolongées en club par des procédures plus complexes en liaison avec une initiation à la programmation (cercles inscrits et ex-inscrits , affichage de résultats .....)
- travail autonome en salle d'informatique, ouverte en libre service et au CDI où la présence d'un ordinateur a été bénéfique à tout le monde .

GEOMETRIE ne constitue pas une panacée qui vise à résoudre tous les problèmes . Il ne s'agit pas de tout faire avec et bien que la plupart des élèves l'ait rapidement maîtrisé , une minorité parmi eux s'en est rapidement désintéressé .

De tels activités entièrement ouvertes reposent la question de la structure même d'une classe (travail en groupe , aprtage du travail , projet précis , périodes de synthèse) .

Enfin deux problèmes importants sont à étudier de près . Il s'agit d'abord de la liaison entre les constructions faites de façon classique sur papier à la règle et au compas et le travail fait sur ordinateur , les élèves ayant tendance rapidement à privilégier ce dernier . Ensuite , le transfert entre les connaissances acquises avec l'ordinateur et le travail classique lors des contrôles par exemple ne s'effectue pas automatiquement même pour des élèves ayant beaucoup cherché et .. trouvé (étant entendu que ces contrôles portaient sur des connaissances de géométrie et non sur le logiciel lui-même) .

Ces deux derniers points méritent des expérimentations plus poussées et les apports des recherches faites dans ce doamaine seraient utiles . Ils mettent en lumière la difficulté qu'il y a pour intégrer l'Informatique dans une classe autrement que sous la forme de l'EAO classique .

Quoiqu'il en soit , la facilité d'utilisation, la puissance même de l'outil , la qualité et la rapidité des figures obtenues ont été l'occasion de recherches motivantes pour les diverses classes touchées . L'éclatement des strauctures de la classe , l'absence d'erreurs pénalisantes et sans retour , le rôle positif de l'enseignant ont accru l'autonomie des élèves .

La possibilité de pouvoir « essayer toutes ses idées » a conduit lorsque les contraintes horaires l'ont permis à structurer le travail des élèves .

A l'occasion de ces travaux , certains concepts non géométriques ont été découverts ou consolidés (variables, liste de coordonnées de points) .

En conclusion, GEOMETRIE mérite de par sa richesse , insoupçonnée lors de sa création , une utilisation plus systématique dans les classes du premier cycle .

III. GEOMETRIE ET LES ENSEIGNANTS

GEOMETRIE a fait l'objet d'une présentation de 6 heures au Mans et de 3 heures à LAVAL au cours des stages consacrés à l'enseignement par situations-problèmes (sans compter toutes les présentations d'Alain Bois au cours de ses stages) .

Il faut noter que le caractère entièrement ouvert du logiciel fonctionnant en mode direct fait qu'il avait incontestablement sa place dans de tels stages .

En ce qui concerne les enseignants , on peut distinguer deux catégories :

Ceux qui , déjà sensibilisés à LOGO ont très vite maîtrisé GEOMETRIE . Il semble donc qu'ils puissent rapidement l'utiliser à condition d'accepter d'enseigner par une méthode d'essais-erreurs liée à des situations-problèmes suffisamment motivantes .

Quant aux autres , qui pour la plupart n'avaient jamais touché à un ordinateur , ils ont été séduits par les possibilités du logiciel . Le nombre volontairement limité de procédures et de concepts informatiques mis en jeu semblent faciliter une prise en main rapide âr des non-initiés , une fois réglés les problèmes d'accès au clavier et de chargement . Dans un second temps , GEOEMETRIE pourrait être l'occasion de s'initier à LOGO . L'idée serait alors de concilier une formation in initiale en informatique et des des appliactions pédagogiques immédiates dans la classe , aspect souvent négligé dans les formations proposées .

Outre les activités propsées aux élèves , il est rapidement apparu que GEOMETRIE avec des moyens simples permet de fabriquer rapidement des imagiciels tout à fait performants (voir travaux d'élèves) sur des sujets divers . D'autres idées comme la construction animée d'une figure ou au contraire la fabrication de la figure à l'envers ont été évoquées .

En conclusion, GEOMETRIE semble un execellent outil pour les élèves qui ont effectué un grand nombre de recherches . Le problème des transferts de connaissances reste malgré tout posé . Quant aux enseignants , que ce soit dans les stages d´A. Bois ou lors des stages IREM , on peut dire qu´ils ont réservé au logiciel un bon accueil . C´est la raison pour laquelle il est maintenant nécessaire de diffuser largement ce logiciel ne serait-ce que pour l´améliorer et le compléter .

Fait au Mans , le 29 Mai 1986 .

ANNEXES : quelques thèmes abordés au collège Alain Fournier en classe de 4ème et 3ème.

- médiatrice d'un segment
- cercle circonscrit
- hauteurs d'un triangle
- bissectrice
- cercle - inscrit - cercle exinscrit
- projection - rapport de projection
- symétrie orthogonale
- relations métriques dans le triangle rectangle

- repérage dans le plan : coordonnées
- la boîte à outils
- les fiches GØ , G1, G2 et les fiches projets

EXEMPLE 1

Construction de la bissectrice - Travail en partie en classe - Prolongement en club

METHODE 1 : à partir des caps construction classique



inconvénient : la bissectrice n'est pas nommée par deux points ce qui pose un problème pour une utilisation ultérieure (cercle inscrit)

METHODE 2 : à partir du triangle isocèle



ici la bissectrice est nommée par 2 points A et I

METHODE 3 Utilisation implicite du losange



APPLICATION : CERCLE INSCRIT



EXEMPLE 2 Projection - Proportionnalité - Thales

Travail effectué avec les deux classes de quatrième (tout le monde a fait au moins une figure dans le temps imparti - 1 h 30 environ - et a effectué les mesures).

TEXTE PROPOSE

Tracer à l'écran deux droites - choisir sur l'une d'elle 4 points les projeter orthogonalement sur la deuxième - compléter le tableau - interpréter.





AB	BC	CD	AD
EF	FG	GH	EH

PROLONGEMENT EN CLUB : la procédure Thales a été fabriquée (uniquemen avec la projection orthogonale) mais la version projection quelconque a permis à ceux qui n'avaient pas bien compris de recommencer et d'envisager un grand nombre de figures.

La réciproque a été évoquée de la façon suivante :

on réalise la figure ci-dessous



QUESTION La tortue étant en F que doit on faire pour l'amener en G

PROLONGEMENT EN CLUB : la procédure créée permet d'envisager un grand nombre de figures.



EXEMPLE 3 Travail effectué en club par un élève à la suite d'activités faites en classe.

Dans un triangle construire le cercle inscrit et les trois cercles exinscrits.



PRINCIPE On pointe 3 points avec le stylo optique et la construction se fait automatiquement.

EXEMPLE 4 Relations métriques dans le triangle rectangle (réalisé avec une classe de 3ème).

TEXTE PROPOSE

Tracer un segment [BC] placer A
tel que ABC soit rectangle en A. Recommencer



- une fois déterminées les valeurs possibles de AB, cette longueur a été choisie pour variable
- le calcul de BM et AM s'est fait d'abord avec calculatrice
  les valeurs obtenues ont permis de piloter la tortue jusqu'au poi
  A trouvé, ceci en mode direct
- la répartition des points obtenus alors que AB augmentait
  de 5 en 5 a donné lieu à des discussions.

PROLONGEMENT EN CLUB : une procédure permettant le tracé automatique a été fabriquée

EXEMPLE 5 : médiatrice
cercle circonscrit





EXEMPLE 6 Symétrie orthogonale

1) construire une droite - construire un point et son symétrique

2) tracer la figure symétrique d'une figure usuelle

Pratiquement tous les élèves sont parvenus en 1 h à tracer le symétr d'un triangle.

Un élève a utilisé la procédure intercercles les autres ont utilisé le projeté orthogonal (qu'ils connaissaient bien) et reporté les distances. L'absence d'une procédure fournissant l'intersection d'ur droite et d'un cercle n'a pas permis de simuler toutes les construct







EXEMPLE 7 : hauteurs d'un triangle





REPERAGE DANS LE PLAN : COORDONNEES

La tortue repère sa position sur l'écran à l'aide de deux nombres, ses <u>COORDONNEES</u>



- Tu peux déplacer la tortue où tu veux sans changer son CAP

- Le crayon peut être levé (LC ou baisser (BC)

→ FPOS [60 100] change les deux coordonnées en même temps.

→ FX - 60 change l'abscisse en faisant un glissement horizontal.

→ FY 90 change l'ordonnée en faisant un glissement vertical.

ORIGINE est un primitif LOGO qui signifie FPOS [0 0]

Pour connaître la position de la tortue il suffit de taper EC POS.
Le premier nombre est l'abscisse, le second est l'ordonnée.

** <u>Pour faire des exercices supplémentaires on pourra utiliser la fiche A6.</u>

LA BOITE A OUTILS

LES MOTS	LEUR ACTION	SYNTAXE
1) POS	Donne la position de la tortue (ses coordonnées)	EC POS
2) FPOS	Place la tortue à un endroit donné	FPOS [40 60]
3) XCOR	Donne l'abscisse de la tortue	EC XCOR
4) YCOR	Donne l'ordonnée de la tortue	EC YCOR
5) FX	Modifie l'abscisse sans toucher l'ordonnée	FX 40
6) FY	Modifie l'ordonnée sans toucher l'abscisse	FY 70
7) X	Donne l'abscisse d'un point	EC X :A
8) Y	Donne l'ordonnée d'un point	EC Y :A
9) AXES	Dessine les axes	AXES
10) CAP	Donne le cap de la tortue	EC CAP
11) FCAP	Fixe le Cap (entre 0 et 360)	FCAP 65
12) VERS	Fixe le cap vers un point	FCAP VERS [40 50] ou FCAP VERS :A
13) POINT	Allume un point sur l'écran	POINT [50 60] ou Point
14) DONNE	Permet de mettre une valeur dans une variable	DONNE "A [60 70]
15) CONFONDUS ?	Précise si deux points sont confondus ou non	EC CONFONDUS ? :A :I
16) ELT ?	Précise si un point A appartient à une droite (BC)	EC ELT ? :A :B :C
17) DEBUT	Permet de repartir à zéro	DEBUT

FICHE G Ø : DES POINTS , DES SEGMENTS , ETC.....

LES PRIMITIFS LOGO UTILES		LES MOTS NOUVEAUX
POS	Position	METS
FPOS	Fixe position	TRACE
CAP	Cap	EFFACE
FCAP	Fixe CAP	LISTE POINTS
DONNE		
POINT		

* COMMENT PLACER UN POINT SUR L'ECRAN ?

On utilise le mot **METS**

Ex : DONNE "A [30 40] A sera le nom du point, puis METS "A
ou DONNE "A STYLO puis METS "A

On peut choisir comme point l'endroit où se trouve la tortue :

DONNE "A POS puis METS "A

Pour connaître les coordonnées de A taper EC :A
A est le nom du point :A désigne le contenu de A c'est-à-dire ses coordonnées.

* CONNAITRE LA LISTE DE TOUS LES POINTS

On utilise le mot **LISTEPOINTS**

Taper EC LISTEPOINTS (au départ cette liste est vide)

* EFFACER UN POINT

On utilise le mot **EFFACE**

Ex : EFFACE "A

* TRACER UN SEGMENT

On utilise le mot **TRACE**

Placer deux points A et B par exemple, taper TRACE :A :B

* PROJETS : Voir fiche exercices FICHE G Ø

* <u>FICHE G Ø</u> : P R O J E T S

1. Tracer un segment
2. Tracer un triangle
3. Tracer un triangle isocèle
4. Tracer un triangle équilatéral
5. Tracer un triangle rectangle
6. Tracer un quadrilatère, un pentagone, un polygone quelconque
7. Tracer un quadrilatère non convexe, croisé
8. Tracer un quadrilatère quelconque et ses diagonales
9. Tracer un rectangle, un losange, un carré
10. Tracer un polygone quelconque et ses diagonales

etc.. tous les projets sont les bienvenus.

<u>LOGO</u> : Faire des mots nouveaux permettant de tracer automatiqueme

1. un triangle
2. un quadrilatère
3. un pentagone
4. un polygone et ses diagonales

etc...

Remarques :
a) l'abscisse doit être entre - 160 et 160
b) l'ordonnée doit être entre - 99 et 99
c) pour faciliter le travail il est souhaitable de préparer le travail sur une feuille quadrillée.

<u>IMPORTANT</u> : **BIEN NOTER CHAQUE PROJET REALISE SUR UNE FEUILLE**

FICHE G1 : DISTANCE, MILIEU, MEDIATRICE, MEDIANES CONSTRUCTIONS DE TRIANGLES

LES MOTS NOUVEAUX

DISTANCE	Donne la distance entre deux points
MILIEU	Donne les coordonnées du milieu d'un segment
VERS	Permet de diriger la tortue vers un point donn

* CONNAITRE LA DISTANCE ENTRE DEUX POINTS

On utilise le mot **DISTANCE**

```
Ex :    EC    DISTANCE    :A        :B
        EC    DISTANCE    [30 60]   [-50 90]
        EC    DISTANCE    :A        POS
        etc.....
```

* CONSTRUIRE LE MILIEU D'UN SEGMENT

On utilise le mot **MILIEU**

```
EX :    EC  MILIEU  :A  :B  donne les coordonnées du milieu du
                            segment  [AB]
        DONNE "I  MILIEU  :A  :B
```

* SE DIRIGER VERS UN POINT

On utilise le mot **VERS**

```
EX :    FCAP    VERS    :A
        FCAP    VERS    [30 60]
```

* PROJETS : voir fiche PROJETS FICHE G1

* **FICHE G1 :** P R O J E T S

1. Tracer un triangle isocèle
2. Tracer un triangle équilatéral
3. Tracer un triangle isocèle dont les angles à la base mesurent 40°, 70°, 50° etc....
4. Tracer un triangle ABC dont le côté $[BC]$ mesure 40, l'angle $\hat{B}$ 50° et langle $\hat{C}$ 60°.

   Fabriquer d'autres exemples.

5. Tracer la médiatrice d'un segment
6. Tracer les médianes d'un triangle
7. Tracer un angle et sa bissectrice
8. Tracer un triangle et ses bissectrices
9. Tracer un triangle et ses médiatrices
10. Pour d'autres idées voir la fiche "autour des quadrilatères".

etc....

LOGO : Faire des mots nouveaux permettant de tracer automatiquement :

1. un segment et sa médiatrice
2. un angle et sa bissectrice
3. un triangle et ses médianes
4. un triangle et ses médiatrices
5. un triangle et ses bissectrices

etc.....

IMPORTANT : **BIEN NOTER CHAQUE PROJET REALISE SUR UNE FEUILLE**

FICHE G2 : INTERSECTION DE DROITES, CERCLES

LES MOTS NOUVEAUX

INTERDROITES	Donne les coordonnées du point d'intersection de deux droites.
PROJETE	Donne les coordonnées du projeté orthogonal d'un point sur une droite.
CERCLE	Permet de tracer un cercle en donnant le centre et le rayon.
DISTANCEDROITE	Donne la distance d'un point à une droite.

*CONNAITRE LES COORDONNEES DU POINT D'INTERSECTION DE DEUX DROITES

On utilise le mot **INTERDROITES**

```
Ex : EC  INTERDROITES  :A   :B  :C  :D
    ou DONNE  "M   INTERDROITES  :A  :B  :C  :D  METS  "M
    ou FPOS  INTERDROITES :A  :B  :C  :D  puis
       DONNE  "M  POS   METS  "M
```

*TRACER UN CERCLE

On utilise le mot **CERCLE**

```
Ex  :  CERCLE   :A   30   A est le Centre,  30 le rayon
    ou CERCLE   :A   DISTANCE  :A   :B
    ou CERCLE   :A   DISTANCEDROITE  :A  :B  :C
```

*CONNAITRE LES COORDONNEES DU PROJETE ORTHOGONAL D'UN POINT SUR 1 DROI

On utilise le mot **PROJETE**

```
Ex  :  EC  PROJETE  :A  :B  :C  donne les coordonnées du projeté
       orthogonal de  A  sur  (BC)
    ou DONNE  "I  PROJETE  :A  :B  :C  METS  "I
    ou FPOS  PROJETE  :A  :B  :C   DONNE  "I  POS  puis  METS  "I
    ou FCAP  VERS  PROJETE  :A  :B  :C
```

*CONNAITRE LA DISTANCE D'UN POINT A UNE DROITE

On utilise le mot **DISTANCEDROITE**

```
Ex  :  EC  DISTANCEDROITE   :A  :B  :C  donne la distance du point
       A  à  la droite  (BC).
```

Utilisation des macro-procédures
dans l'enseignement de
la
Géométrie au Collège

Nous nous sommes rendus compte que si LOGO en sixième s'intégrait bien dans l'enseignement de la Géométrie (voir travaux de J.I Guillot) au contraire nous avions des problèmes quant à son utilisatio dans les classes de $5^{ème}$ et suivantes . C'est dans ce but que nous avons pensé à construire des macro-procédures .

Généralités

1. Qu'est-ce qu'une macro-procédure ? Comme son nom l'indique c'est une «grosse» procédure qui a le statut de primitive dans un micro-monde

2. Qu'est-ce qu'un micro-monde ? D'après Pappert , un micro-monde LOGO est un environnement pédagogique dans lequel on place l'enfant ave des outils LOGO qui doivent lui permettre de se déplacer et de vivre dans ce micro-monde .

3. Quel micro-monde ? A partir de la classe de cinquième , il nous a semblé possible de placer l'élève dans le micro-monde des « constructions » de figures géométriques .

4. Quels outils ? Pour que l'enfant s'approprie les outils du micro-monde «constructions géométriques» , il faut lui fournir :

- des procédures actions : TRACE, CERCLEC, SEGMENT..
- des procédures objets : qui rendent des coordonnées ou des listes de coordonnées ou des nombres

Voyons sur un exemple ce que cela peut donner (vous trouverez d'autres exemples dans le compte-rendu d'activités de J. Delgoulet) :

Un exemple d'utilisation

1. Méthodologie : Dans l'esprit des situations-problèmes , nous fournissons à l'élève une feuille sur laquelle figurent le texte d'u problème et une liste de nouvelles primitives LOGO («boîte à outils»

L'élève doit résoudre le problème posé en utilisant les outils LOGO il peut utiliser le mode direct ou le mode procédural (Le mode procédural lui permettra de découvrir l'intérêt des variables) .

2. Un exemple :

Problème posé : Construis les bissectrices des angles d'un triangle

Selon les objectifs que le professeur fixe à cette situation , o peut envisager différentes « boîtes à outils » .

Si l'on veut que l'élève trace la bissectrice par la méthode traditionnelle , voici les outils que l'on fournira :

- CERCLEC :O :R procédure action qui permet de tracer un cercle de centre O et de rayon R

- INTERCERCLES :O :R :O' :R' Procédure objet qui rend la liste des coordonnées des points d'intersection de deux cercles

Si l'on veut que l'élève construise la bissectrice par la méthode du triangle isocèle , on lui fournira :

- MILIEU :A :B procédure objet qui permet de connaître les coordonnées du milieu de deux points .

On peut envisager de demander à l'élève de construire une procédure BISSECTRICE :A :B :C qui servirait à tracer la bissectrice de ABC , mais aussi celle de BCA et aussi celle de CAB .

Supposons qu'ils ont comme objectif le tracé des bissectrices par la méthode du triangle isocèle .

3. Que va faire l'élève ?

- Demander le tracé d'un triangle : l'élève tape TRIANGLE . L'ordinateur demande à l'élève de placer le stylo optique aux 3 sommets du triangle . Puis il va lui demander de nommer ce triangle . (exemple ABC)

- Tracer la bissectrice de l'angle A par exemple .

Fig 1.

```
1. FPOS :B FCAP VERS :A AV 50
2. FPOS :B FCAP VERS :C AV 50
```

A ce moment-là les élèves se rendent compte qu'il faut nommer les points dont on veut connaître le milieu , aussi il recommence

Fig 2.

```
1. FPOS :B FCAP VERS :A AV 50 NOMME "I
2. FPOS :B FCAP VERS :C AV 50 NOMME "J
3. FPOS MILIEU :I :J NOMME "O

4. TRACE [ O A ]
```

On peut envisager une modification de la syntaxe de la primitive MILIEU , on peut très bien envisager la syntaxe suivante (MILIEU [A B]) .

Maintenant que la bissectrice de l'angle A est tracée , on peut envisager la construction d'une procédure BISSECTRICE sur le modèle du tracé obtenu en mode direct .

```
POUR BISSECTRICE :A :B :C
   FPOS :A FCAP VERS :B DONNE "I POS
   FPOS :B FCAP VERS :C DONNE "J POS
   LC FPOS MILIEU :I :J BC FPOS :B
FIN
```

Aprés avoir vérifié la « qualité » de sa procédure , on peut envisager de lui demander la recherche du point d'intersection des bissectrices , vous envisagerez la suite .....


A
B
C


A
I
O
J
B


A
B
C


A
B


A
B
C
M

Utilisation possible du logiciel GEOMETRIE

Tracés de Coniques

Problème posé : Soit une droite AB et un point F .
1. Tracez un cercle tangent à (AB) et passant par F
2. Dessinez l'ensemble des centres des cercles qui répondent à la première question
3. Tracez l'ensemble des médiatrices des segments [FX_n] tels que X_n soit un point de la droite AB

Outils utilisables : - Les outils généraux du langage LOGO (AV, FPOS, TD ....) plus les nouvelles primitives suivantes :

- MILIEU Liste Liste contient les noms des extrémités du segment
- INTER :A :B :C :D permet de trouver l'intersection de deux droites
- CERCLEC :O :R permet de tracer un cercle dont on connait le centre et la valeur du rayon
- DISTANCE :A :B Rend la distance des deux points A et B
- DISTANCEDROITE :A :B :C Rend la distance du point A à la droite BC

Solutions possibles :

[redacted] 1. Tracé d'un cercle tangent à (AB) et passant par F

11. On place le point F et la droite (AB) avec le stylo optique (Placer si possible A et B en haut et en bas de l'écran)

12. On place la tortue en A ... FPOS :A
On lui demande de se diriger vers B ... FCAP VERS :B
On avance de x pas ..... AV 30
On peut nommer ce point ...... NOMME "M

13. Il faut maintenant trouver le point d'intersection de la médiatrice de [FM] et de la perpendiculaire menée en M à (AB). Voici une méthode possible :

```
FCAP VERS :F AV ( DISTANCE :M :F ) / 2
DONNE "m POS
```
(On utilise DONNE plutôt que NOMME pour éviter d'encombrer la figure avec des lettres)

```
AV x DONNE "m1 POS
FPOS :M FCAP VERS :A
TD 90 AV x DONNE "m2 POS
FPOS INTER :m :m1 :M :m2
NOMME "O     CERCLEC :O DISTANCE :O :F
          ou CERCLEC :O DISTANCEDROITE :O :A :B
```

Cette méthode en mode direct peut paraître longue mais elle va nous permettre d'aller beaucoup plus vite pour les deuxième et troisième questions.

[redacted] 2. Tracé de l'ensemble des centres de cercles

```
POUR CERCLES :F :A :B :pas
      LC FPOS :A FCAP VERS :B AV :pas
      DONNE "M POS
      FPOS :M FCAP VERS :F AV ( DISTANCE :M :F ) / 2
      TD 90
      DONNE "m POS AV x DONNE "m1 POS
      FPOS :M FCAP VERS :A
      TD 90 AV x DONNE "m2 POS
      FPOS INTER :m :m1 :M :m2
      BC POINT POS
      CERCLES :F :A :B :pas + 5 ( par exemple )
FIN
```

Il faudrait peut-être prévoir un test d'arrêt .

[redacted] 3. Tracés des enveloppes de la parabole

```
POUR ENVELOPPES :F :A :B :pas
      SI :pas > DISTANCE :A :B [ STOP ]
      LC FPOS :A FCAP VERS :B AV :pas
      DONNE "M POS
      FPOS :M FCAP VERS :F AV ( DISTANCE :M :F ) / 2
      TD 90
      DONNE "m POS AV 50 DONNE "m1 POS
      DONNE "CAP CAP
      FPOS :M FCAP VERS :A
      TD 90 AV x DONNE "m2 POS
      FPOS INTER :m :m1 :M :m2
      BC FCAP :CAP AV 100 RE 200
      ENVELOPPES :F :A :B :pas + 5
FIN
```




F.






F



Récursivité et Barycentre

Pour rechercher le barycentre de n points affectés des coefficients c_1 ..... c_n, je cherche le barycentre des deux premiers points, puis je cherche le barycentre des (n − 2) points qui restent et du barycentre des deux premiers points affecté du coefficient somme des coefficients des deux premiers points. N'est-ce pas un magnifique exemple de procédure récursive !!!!

Voyons comment on peut traiter celà en LOGO.
Nous utiliserons deux variables :
- #A contiendra la liste des points dont on veut connaître le barycentre (ex : [A B C D E])
- #B contiendra la liste des coefficients (ex : [1 2 5]

```
POUR BARYCENTRE :#A :#B
 SI EGAL? COMPTE :#A 2 [RENDS Barycentre.de.deux.points ]
  DONNE "Bary BARYCENTRE LISTE PREM :#A PREM SP :#A
                         LISTE PREM :#B PREM SP :#B
     RENDS BARYCENTRE MP "Bary SP SP :#A
                      MP SOMME PREM :#B ITEM 2 :#B SP SP :#B
FIN
```

On appelle Bary le barycentre des deux premiers points de la liste #A.

Quand il ne reste plus que deux points, il faut bien calculer les coordonnées du barycentre de deux points.

On place le barycentre au début de la liste #A dont on enlève les deux premiers points,

On remplace dans la liste #B les deux premiers coefficients par la somme des deux premiers coefficients.

Il ne reste plus qu'à écrire la procédure qui permet de calculer les coordonnées du barycentre de deux points.

```
POUR Barycentre.de.deux.points
  RENDS  PH  DIV  SOMME  PROD PREM CHOSE PREM :#A PREM :#B PROD
PREM CHOSE PREM SP :#A PREM SP :#B SOMME PREM :#B PREM SP :#B

                DIV SOMME PROD DER CHOSE PREM :#A PREM :#B PROD  DER
CHOSE PREM SP :#A PREM SP :#B SOMME PREM :#B PREM SP :#B
FIN
```

Quelques explications sur cette procédure :

1. Un point du plan est désigné par une lettre et par ses coordonnées, d'où la nécessité de différencier la lettre A et son contenu :A ou CHOSE "A.

2. les coordonnées du barycentre G de deux points A et B de coordonnées (x_A y_A) et (x_B y_B) et de coefficients c_A et c_B sont données par les formules suivantes :

$$x_G = \frac{c_A \cdot x_A + c_B \cdot x_B}{c_A + c_B} \qquad {}_xG = \frac{c_A \cdot y_A + c_B \cdot y_B}{c_A + c_B}$$

Voyons sur quelques exemples la recherche de barycentres :

- Isobarycentre de 3 points ..... Fig 1

- Barycentre de 3 points A, B, C respectivement affectés des coefficients distance de B à C, distance de A à C, distance de A à B .

- Isobarycentre de 36 points sommets d'un polygone régulier avec tracé de la courbe décrite par les barycentres successivement trouvés par la procédure BARYCENTRE ... Fig 3

- Même figure mais cette fois-ci avec tracés des "lignes de forces" .... Fig 4

- Même figure mais avec plus de points ... Fig 4 bis

- Barycentre de n points sommets d'un polygone régulier avec des coefficients répondant à la propriété suivante :

- $c_x = x^2$ G2

Fig 5 - $c_x = (n - x)^2$ G1



Fig 1

Fig 2





Fig. 3



Fig 4.



Fig. 4 bis



Fig 5.

Titre : Logo et géométrie

Auteurs : Alain BOIS, Jacques DELGOULET

Niveau : 4e – 3e

Date : 2007

Mots clés :

IREM des Pays de la Loire
Centre de Nantes
2, rue de la Houssière – BP 92208
44322 NANTES CEDEX 03

Format : A4 *44 pages* *Prix : 3 €*